U. S. DEPARTMENT OF AGRICULTURE
SOIL CONSERVATION SERVICE
IDAHO

IDAHO TECHNICAL NOTE NO. 6

DESIGN OF DUMPED ROCK RIPRAP STREAM CHANNEL STABILIZATION

INTRODUCTION

In the past few years there have been a number of methods proposed for designing rock riprap for channel or river stabilization. Most of these methods are rather complex, using tractive force to predict movement of rock and soil particles when subjected to water and gravity. Recommended rock sizes found by the proposed methods vary widely and questions have been raised regarding their validity.

In consultation with Engineers at the West National Technical Center, it has been determined that the method presented in Chapter 6 of the publication, "Engineering Design Standards, Far West States", produces rock sizes that are reasonable and accurate. The procedure has been widely used and accepted for over 30 years. Therefore, this method will be used for design of rock riprap for channel stabilization until further notice.

While it is a simple, straight-forward method of solution to determine the rock size, D_{75}, for which 75% of the rock is finer, the Far West States (FWS) method does not give recommendations to determine the rock gradation, the sizes needed for stream bottom protection, layer thickness, or filter design. This technical note will address those needs, along with additional discussion of the basic equation.

DESIGN PROCEDURE

Riprap for Channel Banks. The FWS method uses a single equation to deal with the variables for riprap on stream or canal banks:

$$D_{75} = \frac{3.5}{CK} wDS \qquad \text{for Channel Banks}$$

where: D_{75} = Size of rock for which 75% is finer in the gradation, in inches

w = specific weight of water, usually 62.4 #/cu.ft.
D = Depth of flow in stream, in feet.
S = Channel slope or gradient, in ft/ft.
C = A coefficient relating to curvature in the stream.
K = A coefficient relating to steepness of bank slope.

The coefficient, C, is based on the ratio of the radius of curvature of the stream, (CR), to the water surface width, (WSW), so it is necessary for the user to make field determination of these values. The coefficient varies from 0.6 for a curve ratio of 4 to 6, up to 1.0 for a straight channel. If the computed ratio for a particular project is less than 4, the designer should consider some modification of the channel alignment to eliminate such sharp curves.

CR/WSW	C
4 - 6	0.60
6 - 9	0.75
9 - 12	0.90
Straight Channel	1.00

The coefficient, K, ranges from 0.5 for a 1.5:1 sideslope to 0.87 for 3:1 sideslope. No values are given for steeper or flatter slopes. Slopes steeper than 1.5:1 are not recommended. If slopes flatter than 3:1 are desired, it would be conservative to use the K-value for 3:1 slopes.

Bankslope	K
1.5:1	0.50
1.75:1	0.63
2.0:1	0.72
2.5:1	0.80
3.0:1	0.87

Riprap for Channel Bottom. For designing riprap for the bottom of a stream or canal, the bank equation can be modified as follows:

$$D_{75} = 2.5 \frac{wDS}{C} \qquad \text{for Channel Bottom}$$

Obviously, the K-value does not apply and can be omitted. It is assumed that the channel bed shear is about 40% less than the bank shear.

Rock Gradation. The riprap gradation should yield a smooth size distribution curve from largest to smallest size to be installed. For well graded riprap, the interstices between larger rocks are filled with smaller rocks; thus the potential for leaching filter or base material through the rock is reduced. As a practical matter, the designer should give a range of acceptable sizes for each percentage level. This gives the supplier some latitude in finding suitable materials for the rock riprap. The following relationships are used by the US Bureau of Reclamation and have proven satisfactory for most SCS work.

RECOMMENDED RIPRAP GRADATION

% Finer by Weight	Minimum Size	Maximum Size
D_{100}	1.33 x D_{75}	2.0 x D_{75}
D_{75}	1.0 x D_{75}	1.67 x D_{75}
D_{50}	0.67 x D_{75}	1.17 x D_{75}
D_{25}	0.33 x D_{75}	0.77 x D_{75}
D_0	None	0.33 x D_{75}

Filter Design. Leaching is the process by which the finer base materials beneath the riprap are picked up and carried away by the turbulence that penetrates the interstices of the riprap. As the base materials disappear, the rock armor collapses into the cavity and eventual failure of the riprap will occur. Leaching is reduced to a negligible rate by using a properly designed filter under the riprap or by making the riprap layer thick enough and with fine enough

interstices to keep erosive currents away from underlying soil.

Filter design should be based on Soil Conservation Service Soil Mechanics Note No. 1, (Revised January 1986). The designer should first evaluate the base material in relationship to the proposed rock riprap using the filter design criteria to determine whether a filter is required. It is strongly recommended that the designer plot all gradations on Standard Form ID-ENG-10, SCS-ENG-353, or SCS-ENG-130 to visually see the relationships of the base material, filters and rock riprap.

If a filter is needed, proposed gravel filter gradations should be determined and plotted to see how well they will meet the criteria. These proposed filters should be materials that are locally available, such as pit-run gravels from gravel sources, or processed sands and gravels from local concrete suppliers.

If the filter requirements between the riprap and the base soils still cannot be met, or if local materials are not suitable, consideration should be given to the use of non-woven filter cloth. The cloth may be used either with or without gravel filters but its characteristics must be determined from West NTC Technical Note No. W-30 and manufacturers literature. In determining filtration characteristics, Technical Note No. W-30 states simply that the apparent opening size (AOS) of non-woven geotextiles must not exceed the No. 40 sieve. The AOS of the selected geotextile should be plotted on the gradation chart and checked with the filter criteria for suitability with the materials that will be on either side of it.

In a few cases, a combination of one size filter cloth, gravels, and a different size filter cloth may be needed to meet all criteria. Graded gravel filters should have a minimum thickness of 8 inches. In many cases, 12 inches is more desirable to ensure adequate separation of the base material from the rock riprap.

Without reviewing the contents of SM-1, the basic steps for determining gravel filter gradation limits are summarized here. Some comments are included to give guidance for routine application of the procedure.

1. Determine the unified soil classification and gradation curve of the base material. This may be done by field evaluation methods for most smaller projects. If the soil contains gravels (particles larger than No. 4 sieve), a typical sample should be sieved, and an adjusted distribution curve prepared as described in SM-1 to find the percentage passing the No. 200 sieve.

2. Determine gravel filter criteria from the following table.

RECOMMENDED GRAVEL FILTER CRITERIA

Base Soil Description w/ % Finer than 200 Sieve	Filter Criteria (See Notes Below Table)
Fine silts and clays, more than 85% finer	$D_{15} < 9 \times D_{85}$, but not less than 0.2mm
Sands, silts, clays silty & clayey sands 40 to 85% finer	$D_{15} < 0.7$ mm
Silty & clayey sands and gravels 15 to 39% finer	$D_{15} < ((40 - A)/(40 - 15) \times (4 \times D_{85} - 0.7) + 0.7)$
Sands and gravels less than 15% finer	$D_{15} < 4 \times D_{85}$

NOTES: Additional Filter Criteria for all soils:

1. Filters shall have a maximum particle size of of 3 inches and less than 5% passing the No. 200 sieve. PI of fines shall be zero.
2. D_{15} of filters shall not be less than 4 x D_{15} of the base material and not less than 0.1 mm.
3. A = percent passing the No. 200 sieve after any regrading.
4. For silty and clayey sands and gravels, with 15 to 40 % passing the No. 200 sieve, if 4 x D_{85} is less than 0.7 mm, use 0.7 mm in the equation.

Thickness of Riprap Lining. Thickness of riprap linings should be determined by experience in the area, type of rock, construction techniques, size of channel, etc. The following criteria is a minimum for most projects in Idaho.

1. Thickness at base of slope equals twice the D_{75} size if a filter is used, and triple the D_{75} if a filter is not used.

2. Thickness at top of bank equals 1.5 x the D_{75} size.

3. For channel bottom riprap, thickness equals 1.5 times the D_{100}.

Toe Protection. When the channel bottom is not riprapped, the toe of the bank protection requires special attention. Many riprap projects fail because the flood water undermines the toe of the slope.

The recommended method of toe protection is to excavate a trench a minimum of 3 feet deep and 4 to 8 feet wide and fill it with rock as a foundation for the slope rock.

If it is not possible to excavate the trench, then a layer of rock should be placed at the toe of the slope, with a range of dimensions as follows: Layer thickness should be a minimum of $3 \times D_{75}$. After placement of the slope riprap on top of the toe protection, the top of the toe protection should still project out a minimum of 4 feet into the stream.

A typical drawing sheet is attached showing these alternate toe protection details.

Transitions. The upstream and downstream end of any riprap project requires special attention to prevent moving water from attacking the base material behind the riprap blanket. Field investigation should establish beginning and ending points at locations where stream currents are moving relatively straight and uniform in a direction parallel to the stream channel banks for some distance upstream. Site conditions should give some assurance that the stream meander pattern will not change in the foreseeable future.

In addition to careful site selection, construction should include a thickened rock section at the beginning and end of the riprapped section. As a minimum at both ends, the bank and channel bottom should be excavated to accommodate a layer thickness that is double the designed thickness. This thickness should extend for a distance of about 6 feet on each end.

Dumped Rock Riprap
Stream Channel Stabilization

EXAMPLE PROBLEM

A farmer has requested technical assistance in designing a rock riprap bank protection for a curved section of river in his property. The field investigation includes a topographic survey of the section of river involved and a sieve analysis of the typical natural streambank material. From the data collected and some hydrology analysis the following information is found:

1. Water depth at design flow = 6.0 feet
2. Water surface width at design flow = 70 feet.
3. Channel slope in vicinity of curve = 0.0040 feet/foot.
4. Radius of curvature for reach = 600 feet.
5. Gradation of existing material is plotted on attached ID-ENG-10.

We need to find the following:

1. Size of rock riprap needed for bank protection
2. Whether a filter is needed.
3. If a filter is needed, design the filter.
4. Determine the thickness of the rock and the filter.

Procedure:

1. Use Far West States, Engineering Design Standards procedure, HP-41CV calculator program, "ROCK2", to find the required D75.

2. $$D75 = \frac{3.5\ (62.4)(6.0)(.0040)}{(.75)(.72)} = 9.71 \text{ inches}$$

3. Determine what safety factor is needed. Use 1.25.
   9.71 x 1.25 = 12.1 inches

   Use 12 inches for design.

4. Find gradation of the rock riprap using recommendations of this technical note.

	Min. Size	Max. Size
D100	16.0	24.0
D75	12.0	20.0
D50	8.0	14.0
D25	4.0	9.2
D0	-----	4.0

5. Plot gradation of rock on ID-ENG-10 and compare with the native base material.

   a. Must regrade the base material in accordance with SM-1
      (1) Ratio for adjustment = 100/88 = 1.14
      (2) #16 sieve = 69% x 1.14 = 76%
      (3) #40 sieve = 50% x 1.14 = 57%
      (4) #100 sieve = 30% x 1.14 = 34%
      (5) #200 sieve = 23% x 1.14 = 26%

   b. Check criteria for filter from table

      $$D15 < \frac{(40 - 25)}{(40 - 15)} \times ((4 \times 2) - 0.7) + 0.7$$

      $$D15 < (0.6 \times 7.3) + 0.7 = 5.08 \text{ mm}$$

      D15 of the rock riprap is > 75 mm so a filter is needed.

6. A local source has gravel available with a gradation as shown on Curve A on the attached ID-ENG-10. Check its suitability as a filter for the base material. It must meet the following criteria.

   $$D15 < 4 \times D85 = 4 \times 2\text{mm} = 8 \text{ mm} \quad \text{OK.}$$

7. Check the proposed filter material as a base material against the rock riprap.

   $$D15 < 4 \times D85 = 4 \times 1.5\text{inches} = 6 \text{ inches} \quad \text{OK.}$$

8. Find layer thickness of riprap.

   a. Thickness at base of slope = D75 x 2 = 24 inches.

   b. Thickness at top of slope = D75 x 1.5 = 18 inch.

   c. Toe protection thickness = D75 x 3 = 36 inches.

9. Filter thickness minimum of 10 inches.

U. S. Department of Agriculture
Soil Conservation Service

ID-ENG- 10
JUNE, 1985

ROCK RIPRAP, FILTER OR BEDDING GRADATION

PROJECT EXAMPLE PROBLEM SCD _____

BY KDL DATE 11/20/91





* See Alternate Detail

TYPICAL CROSS-SECTION (not to scale)



ALTERNATE TOE CROSS-SECTION (not to scale)

EXAMPLE PROBLEM

STREAM BANK STABILIZATION

TYPICAL ROCK RIPRAP

U. S. DEPARTMENT OF AGRICULTURE SOIL CONSERVATION SERVICE

	Date		
Designed		Approved by	
Drawn		Title	
Traced		Title	
Checked		Sheet No. of	Drawing No.

PROGRAM DESCRIPTION I

Program Title ROCK RIPRAP GRADATION FOR STREAM CHANNEL STABILIZATION

Contributor's Name Karl D. Larson

Address

City **State/Country** **Zip Code**

Program Description, Equations, Variables

Using the procedures from Far West States-Engineering Design Standards, this method calculates the gradation of rock riprap needed for protection of channel banks and bottoms. For bank protection, user must input values for curves, channel width, side slopes to be used. Gradation information used in calculating the required range of rock sizes comes from Bureau of Rec., and Corps of Engineers recommendations.

Necessary Accessories Printer

Operating Limits and Warnings Calculator in User Mode

Reference(s)

USER INSTRUCTIONS

SIZE: (HP-41C)

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	XEQ "ROCK2"			SP. WT. FL=
2	Specific Weight of Fluid Usually62.4		R/S	H2O DEPTH=
3	Water depth in Channel, feet		R/S	Chan Slope =
4	Slope of channel, feet		R/S	ROCK Z =
5	Horizontal component of bank slope Calculates and prints K-value		R/S	CURVE RADIUS=
6	Radius of curve of channel, feet		R/S	WAT SUR WID =
7	Width of water surface, feet Calculates and prints, C-value		R/S	
8	If user wants to find riprap of Bank Protection, Press A If user wants to find riprap for channel bottom protection, Press B (BE SURE CALCULATOR IS IN USER MODE)			
9	Calculates and prints D75 rock size			SAFE FACT =
10	Input Desired Safety Factor		R/S	D75 DESIGN =
11	Input Desired D75 for Design Calc. Calculates envelope curve for gradation of the Rock Riprap.		R/S	

PROGRAM DESCRIPTION II

Sample Problem (Sketch if Desired)

Given Stream Channel Data at location needing protection::

Design Water Depth = 6.0 feet; Channel slope = 0.0040 ft/ft; Curve radius = 600 feet; Water Surface Width = 70 feet; Slope of bank = 2:1; Specific gravity of water = 62.4 #/cu.ft.

Find:

Required rock riprap gradation for protection of both bank and channel bottom.

SOLUTION:

ROCK RIPRAP DESIGN - FAR WEST STATES DESIGN STDS

SP WT FL=62.4
H2O DEPTH=6.0
CHAN SLOPE=0.0040
ROCK Z =2.00
CONSTANT K=0.72
CURVE RADIUS=600.
WAT SUR WID=70.
RATIO CR:WSW=8.57
CONSTANT C=0.75

BANK ROCK
ROCK D75=9.71
SAFE FACT=1.25
NEW ROCK D75=12.1

D75 DESIGN=12.0

GRADATION
D100 MAX=24.0
D100 MIN=16.0
D75 MAX=20.0
D75 MIN=12.0
D50 MAX=14.0
D50 MIN=8.0
D25 MAX=9.2
D25 MIN=4.0
D0 MAX =4.0

BOTTOM ROCK
ROCK D75=5.0
SAFE FACT=1.3
NEW ROCK D75=6.2

D75 DESIGN=6.5

GRADATION
D100 MAX=13.0
D100 MIN=8.7
D75 MAX=10.8
D75 MIN=6.5
D50 MAX=7.6
D50 MIN=4.3
D25 MAX=5.0
D25 MIN=2.2
D0 MAX =2.2

Comments

PROGRAM LISTING

☐ 67 ☐ 97 ☐ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
	01♦LBL "ROCK2"		
	02 SF 12		
	03 "ROCK RIPRAP"		
	04 AVIEW		
	05 "DESIGN - FAR"		
	06 AVIEW		
	07 "WEST STATES"		
	08 AVIEW		
	09 "DESIGN STDS"		
	10 AVIEW		
	11 CF 12		
	12 CLRG		
	13 ADV		
	14 FIX 1		
	15 "SP WT FL="		
	16 PROMPT		
	17 STO 00		
	18 XEQ 01		
	19 "H2O DEPTH="		
	20 PROMPT		
	21 STO 01		
	22 XEQ 01		
	23 FIX 4		
	24 "CHAN SLOPE="		
	25 PROMPT		
	26 STO 02		
	27 XEQ 01		
	28 FIX 2		
	29 "ROCK Z ="		
	30 PROMPT		
	31 STO 10		
	32 XEQ 01		
	33 RCL 10		
	34 1.63		
	35 X<=Y?		
	36 GTO 05		
	37 .5		
	38 STO 09		
	39 GTO 09		
	40♦LBL 05		
	41 RCL 10		
	42 1.87		
	43 X<=Y?		
	44 GTO 06		
	45 .63		
	46 STO 09		
	47 GTO 09		
	48♦LBL 06		
	49 RCL 10		
	50 2.26		
	51 X<=Y?		
	52 GTO 07		
	53 .72		
	54 STO 09		
	55 GTO 09		
	56♦LBL 07		
	57 RCL 10		
	58 2.76		
	59 X<=Y?		
	60 GTO 08		
	61 .8		
	62 STO 09		
	63 GTO 09		
	64♦LBL 08		
	65 RCL 10		
	66 .87		
	67 STO 09		
	68 GTO 09		
	69♦LBL 09		
	70 "CONSTANT K="		
	71 RCL 09		
	72 STO 03		
	73 XEQ 01		
	74 FIX 0		
	75 "CURVE RADIUS="		
	76 PROMPT		
	77 STO 04		
	78 XEQ 01		
	79 "WAT SUR WID="		
	80 PROMPT		
	81 XEQ 01		
	82 RCL 04		
	83 /		
	84 1/X		
	85 FIX 2		
	86 "RATIO CR:WSW="		
	87 STO 11		
	88 XEQ 01		
	89 RCL 11		
	90 6.1		
	91 X<=Y?		
	92 GTO 02		
	93 0.6		
	94 STO 12		
	95 GTO 10		
	96♦LBL 02		
	97 RCL 11		
	98 9.1		
	99 X<=Y?		
	100 GTO 03		
	101 .75		
	102 STO 12		
	103 GTO 10		
	104♦LBL 03		
	105 RCL 11		
	106 12.1		
	107 X<=Y?		
	108 GTO 04		
	109 .9		
	110 STO 12		
	111 GTO 10		
	112♦LBL 04		
	113 RCL 11		
	114 1		
	115 STO 12		
	116 GTO 10		
	117♦LBL 10		
	118 "CONSTANT C="		
	119 RCL 12		
	120 STO 06		
	121 XEQ 01		
	122 STOP		
	123♦LBL A		
	124 ADV		
	125 SF 12		
	126 "BANK ROCK"		
	127 PRA		
	128 CF 12		
	129 RCL 00		
	130 3.5		
	131 *		
	132 RCL 01		
	133 *		
	134 RCL 02		
	135 *		
	136 RCL 03		
	137 /		
	138 RCL 06		
	139 /		
	140 STO 07		
	141 "ROCK D75="		
	142 XEQ 01		
	143 "SAFE FACT="		
	144 PROMPT		
	145 XEQ 01		
	146 RCL 07		
	147 *		
	148 STO 05		
	149 FIX 1		
	150 "NEW ROCK D75="		
	151 XEQ 01		
	152 GTO C		
	153♦LBL B		
	154 ADV		
	155 SF 12		
	156 "BOTTOM ROCK"		
	157 PRA		
	158 CF 12		
	159 RCL 00		
	160 2.5		
	161 *		
	162 RCL 01		
	163 *		
	164 RCL 02		
	165 *		
	166 RCL 06		
	167 /		
	168 STO 13		
	169 "ROCK D75="		
	170 XEQ 01		
	171 "SAFE FACT="		
	172 PROMPT		
	173 XEQ 01		
	174 RCL 13		
	175 *		
	176 STO 05		
	177 FIX 1		
	178 "NEW ROCK D75="		
	179 XEQ 01		
	180 GTO C		
	181♦LBL C		
	182 ADV		
	183 "D75 DESIGN="		
	184 PROMPT		
	185 STO 05		
	186 XEQ 01		
	187 ADV		
	188 SF 12		

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.

PROGRAM LISTING

☐ 67 ☐ 97 ☐ 41C

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
	189 "GRADATION"		
	190 PRA		
	191 CF 12		
	192 RCL 05		
	193 1.5		
	194 /		
	195 STO 05		
	196 RCL 05		
	197 3		
	198 *		
	199 "D100 MAX="		
	200 XEQ 01		
	201 RCL 05		
	202 2.		
	203 *		
	204 "D100 MIN="		
	205 XEQ 01		
	206 RCL 05		
	207 2.5		
	208 *		
	209 "D75 MAX="		
	210 XEQ 01		
	211 RCL 05		
	212 1.5		
	213 *		
	214 "D75 MIN="		
	215 XEQ 01		
	216 RCL 05		
	217 1.75		
	218 *		
	219 "D50 MAX="		
	220 XEQ 01		
	221 RCL 05		
	222 "D50 MIN="		
	223 XEQ 01		
	224 RCL 05		
	225 1.15		
	226 *		
	227 "D25 MAX="		
	228 XEQ 01		
	229 RCL 05		
	230 .5		
	231 *		
	232 STO 08		
	233 "D25 MIN="		
	234 XEQ 01		
	235 RCL 08		
50			

STEP/ LINE	KEY ENTRY	KEY CODE (67/97 only)	COMMENTS
	236 "D0 MAX ="		
	237 XEQ 01		
	238 STOP		
	239*LBL 01		
	240 ARCL X		
	241 AVIEW		
	242 RTN		
	243 .END.		
70			
80			
90			
00			

Note: Refer to "HP-41C OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for specific information on keystrokes. The Function Index is found at the very back of the Handbook. Refer to Appendix E in 67 or 97 "OWNER'S HANDBOOK AND PROGRAMMING GUIDE" for exact keystrokes.